


БГТУ, ФИТ, ПОИТ, 3 семестр, Языки программирования

Введение в язык Ассемблер

1. Вывод в консоль



Синтаксис:

```
BOOL SetConsoleTitle(    // устанавливает заголовок для консольного окна
    LPCTSTR lpConsoleTitle // 32-разрядный указатель на строку lpConsoleTitle –
                           // заголовок консольного окна
);
```

Необходимая библиотека: Kernel32.lib

Соглашение о вызовах: stdcall

Возвращаемые значения: { 0 – функция завершается с ошибкой;
иначе – функция завершается успешно.

```

ExitProcess      PROTO :DWORD ; прототип функции
SetConsoleTitleA PROTO :DWORD ; установить заголовок окна консоли (ANSI)
GetStdHandle     PROTO :DWORD ; получить handle вывода на консоль
WriteConsoleA    PROTO :DWORD, :DWORD, :DWORD, :DWORD, :DWORD ; вывод на консоль

.stack 4096      ; сегмент стека объемом 4096
.const          ; сегмент констант

consoletitle db 'SMW Consol',0
helloworld db 'Hello, World!!!!',0


.data          ; сегмент данных
consolehandle dd 0h ; handle консоли
.code         ; сегмент кода
main PROC     ; начало процедуры

push offset consoletitle ; параметр SetConsoleTitleA: адрес строки
call SetConsoleTitleA    ; вызов SetConsoleTitleA

push -11 ; -11 - handle для стандартного вывода
call GetStdHandle ; получить handle->eax
mov consolehandle, eax

push 0 ; можно 0
push 0 ; можно 0
push sizeof helloworld ; количество выводимых байт
push offset helloworld ; адрес выводимой строки
push consolehandle ; handle для вывода
call WriteConsoleA ; вывести на консоль

```



Синтаксис:

```

HANDLE GetStdHandle(          // извлекает дескриптор потока ввода-вывода
    DWORD nStdHandle          // ввод, вывод или ошибка
);

```

Handle стандартного потока **ввода** -10
 Handle стандартного потока **вывода** -11
 Handle потока сообщений об ошибках "**ошибок**" -12

Синтаксис:

```

BOOL WriteConsole(            // выводит символьную строку в консоль
    HANDLE hConsoleOutput,    // дескриптор (Handle)
    CONST VOID * lpBuffer,     // указатель на строку вывода
    DWORD nNumberOfCharsToWrite, // число выводимых символов
    LPDWORD lpNumberOfCharsWritten, // возвращает число выведенных символов
    LPVOID lpReserved          // зарезервировано
);

```

2. Процедура вывода

```
.586                                ; система команд (процессор Pentium)
.model flat,stdcall                 ; модель памяти, соглашение о вызовах

includelib kernel32.lib            ; компоновщик: компоновать с kernel32.lib
ExitProcess     PROTO  :DWORD      ; прототип функции

includelib msvcrt.lib              ; библиотека времени исполнения C
system          PROTO  C :DWORD     ; вывод cmd-команды

.stack 4096                        ; сегмент стека объемом 4096
.const                                                  ; сегмент констант
consoletitle    db 'SMW Consol',0
str_helloworld  db 'Hello, World!!!!',10,0
str_pause       db 'pause',0
.data                                                    ; сегмент данных

.code                                                    ; сегмент кода
main PROC                                              ; начало процедуры

push  offset consoletitle ; заголовок окна консоли
push  offset str_helloworld ; выводимый текст
call  printconsole        ; вызов процедуры

push  offset consoletitle ; заголовок окна консоли
push  offset str_helloworld ; выводимый текст
call  printconsole        ; вызов процедуры

push offset str_pause      ; адрес выводимой cmd-команды
call system                ; system("pause");

push  0                    ; код возврата процесса (параметр ExitProcess )
call  ExitProcess          ; так должен заканчиваться любой процесс Windows

main ENDP                ; конец процедуры
```

```

; -----
;                               Вывод сообщения в консоль
; -----
include lib kernel32.lib
SetConsoleTitleA  PROTO  :DWORD  ; установить заголовок окна консоли (ANSI)
GetStdHandle      PROTO  :DWORD  ; получить handle вывода на консоль
WriteConsoleA     PROTO  :DWORD,:DWORD,:DWORD,:DWORD,:DWORD ; вывод на консоль

printconsole PROC uses eax ebx ecx edi esi,
                pstr: dword, ptitle: dword

    push ptitle          ; параметр SetConsoleTitleA: адрес строки
    call SetConsoleTitleA ; вызов SetConsoleTitleA

    push -11             ; -11 - handle для стандартного вывода
    call GetStdHandle     ; получить handle->eax


    mov esi, pstr         ; подсчет количества
    mov edi, -1          ; символов (до 0h)
count:              ; в выводимой
    inc edi              ; на консоль строке
    cmp byte ptr [esi + edi],0 ;
    jne count           ; количество символов ->edi

    push 0               ; можно 0
    push 0               ; можно 0
    push edi
    push pstr            ; адрес выводимой строки
    push eax             ; handle для вывода
    call WriteConsoleA   ; вывести на консоль

    ret
printconsole ENDP

end main                ; конец модуля, main - точка входа

```



3. Процедура преобразования числа в символы

```


; -----
;                               Преобразование числа в строку
; -----
int_to_char  PROC uses eax ebx ecx edi esi,
                pstr: dword,          ; адрес строки-результата
                intfield: dword       ; преобразуемое число

    mov edi, pstr                ; адрес результата -> edi
    mov esi, 0                   ; количество символов в результате
    mov eax, intfield            ; число -> eax
    cdq                         ; знак распространили на с eax на edx
    mov ebx, 10                  ; десятичная система счисления
    idiv ebx                     ; eax = eax/ebx, остаток->edx
    test eax, 80000000h          ; результат отрицательный ?
    jz plus                      ; если положительный на plus
    neg eax                      ; eax = - eax
    neg edx                      ; edx = -edx
    mov cl, '-'                  ; первый символ результата '-'
    mov [edi], cl                ; первый символ результата '-'
    inc edi                      ; ++edi
plus:                          ; цикл разложения на степени 10
    push dx                      ; остаток -> стек
    inc esi                      ; ++esi
    test eax, eax                ; eax == 0?
    jz fin                      ; если да то на fin
    cdq                         ; знак распространили на с eax на edx
    idiv ebx                     ; eax = eax/ebx, остаток->edx
    jmp plus                    ; переход на plus
fin:
    mov ecx, esi                ; количество не 0вых остатков = количеству символов в результате
write:                          ; цикл записи результата
    pop bx                      ; остаток из стека ->bx
    add bl, '0'                  ; сформировали символ в bl
    mov [edi], bl                ; bl-> в результат
    inc edi                      ; edi++
    loop write                   ; if (--ecx) > 0 goto write


    ret
int_to_char  ENDP
; -----

```

CDQ



DIV/IDIV



```

.586                                ; система команд (процессор Pentium)
.model flat, stdcall                ; модель памяти, соглашение о вызовах
includelib kernel32.lib             ; компоновщик: компоновать с kernel32.lib
ExitProcess      PROTO :DWORD       ; прототип функции
includelib msvcrt.lib               ; библиотека времени исполнения C
system           PROTO C :DWORD     ; вывод cmd-команды
.stack 4096                          ; сегмент стека объемом 4096
.const           ; сегмент констант
consoletitle    db 'int_to_char',0
str_pause       db 'pause',0
.data           ; сегмент данных
result1         byte 40 dup(0)
                byte 10
result2         byte 40 dup(0)
.code           ; сегмент кода
main PROC      ; начало процедуры

    push -77777777                    ; исходное число
    push offset result1               ; место для результата
    call int_to_char                  ; вызов процедуры преобразования

    push offset consoletitle          ; заголовок окна консоли
    push offset result1               ; выводимый текст
    call printconsole                  ; вызов процедуры


    push 123456789                    ; исходное число
    push offset result2               ; место для результата
    call int_to_char                  ; вызов процедуры преобразования

    push offset consoletitle          ; заголовок окна консоли
    push offset (result2-1)           ; выводимый текст
    call printconsole                  ; вызов процедуры



    push offset str_pause              ; адрес выводимой cmd-команды
    call system                        ; system("pause");





    push 0                            ; код возврата процесса (параметр ExitProcess )
    call ExitProcess                  ; так должен заканчиваться любой процесс Windows
main ENDP                            ; конец процедуры




```









4. Создание статической библиотеки











5. Вызов функции из C++

The screenshot displays a C++ development environment with three main components:

- Code Editor:** Contains the source code for `LP_cpp26callib.cpp`.
 - Line 4: `#include "stdafx.h"`
 - Line 6: `extern "C"` (highlighted with a red box)
 - Line 8: `{`
 - Line 9: `int __stdcall asm241(int x);` (highlighted with a red box)
 - Line 10: `};`
 - Line 11: `int _tmain(int argc, _TCHAR* argv[])`
 - Line 12: `{`
 - Line 14: `int x = asm241(7777);` (highlighted with a red box)
 - Line 15: `return 0;`
 - Line 16: `}`
- Solution Explorer (Обозреватель решений):** Located on the right, it shows the project structure for "Решение 'LP_asm26lib'".
 - Project files: `LP_asm26calllib`, `LP_asm26lib`, and `LP_cpp26callib` (highlighted with a red box).
 - External dependencies (Внешние зависимости): `stdafx.h`, `targetver.h`.
 - Source files (Файлы исходного кода): `LP_cpp26callib.cpp` (highlighted with a red box), `stdafx.cpp`.
 - Resource files (Файлы ресурсов): `ReadMe.txt`.
- Variable Watch Window (Контрольные значения 1):** Located at the bottom, it shows the value of the variable `x` as `-7777` (highlighted with a red box).